NAG Library Routine Document

G05MEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05MEF generates a vector of pseudorandom integers, each from a discrete Poisson distribution with differing parameter λ.

2 Specification

```
SUBROUTINE G05MEF(M, VLAMDA, X, IGEN, ISEED, IFAIL)
INTEGER           M, X(M), IGEN, ISEED(4), IFAIL
double precision  VLAMDA(M)
```

3 Description

G05MEF generates m integers x_j, each from a discrete Poisson distribution with mean λ_j, where the probability of $x_j = I$ is

$$P(x_j = I) = \frac{\lambda_j^I \times e^{-\lambda_j}}{I!}, \qquad I = 0, 1, \ldots,$$

where

$$0 \leq \lambda_j, \qquad j = 1, 2, \ldots, m.$$

The methods used by this routine have low set up times and are designed for efficient use when the value of the parameter λ changes during the simulation. For large samples from a distribution with fixed λ using G05MKF to set up and use a reference vector may be more efficient.

When $\lambda < 7.5$ the product of uniforms method is used, see for example Dagpunar (1988). For larger values of λ an envelope rejection method is used with a target distribution:

$$f(x) = \tfrac{1}{3} \qquad \text{if } |x| \leq 1,$$

$$f(x) = \tfrac{1}{3}|x|^{-3} \quad \text{otherwise.}$$

This distribution is generated using a ratio of uniforms method. A similar approach has also been suggested by Ahrens and Dieter (1989). The basic method is combined with quick acceptance and rejection tests given by Maclaren (1990). For values of $\lambda \geq 87$ Stirling's approximation is used in the computation of the Poisson distribution function, otherwise tables of factorials are used as suggested by Maclaren (1990).

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MEF.

4 References

Ahrens J H and Dieter U (1989) A convenient sampling method with bounded computation times for Poisson distributions *Amer. J. Math. Management Sci.* 1–13

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Maclaren N M (1990) A Poisson random number generator *Personal Communication*

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of Poisson distributions for which pseudorandom variates are required.

Constraint: M ≥ 1.

2: VLAMDA(M) – ***double precision*** array *Input*

On entry: the means, λ_j, for $j = 1, 2, \ldots,$ M, of the Poisson distributions.

Constraint: $0.0 \leq$ VLAMDA$(j) \leq MAXINT/2$, where $MAXINT$ is the largest integer representable on the machine (see X02BBF).

3: X(M) – INTEGER array *Output*

On exit: the m pseudorandom numbers from the specified Poisson distributions.

4: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

5: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, M < 1.

IFAIL $= 2$

On entry, VLAMDA$(j) < 0.0$ for at least one value of j.

IFAIL $= 3$

On entry, $2 \times$ VLAMDA$(j) > MAXINT$ for at least one value of j.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints ten pseudorandom integers from five Poisson distributions with means $\lambda_1 = 0.5$, $\lambda_2 = 5$, $\lambda_3 = 10$, $\lambda_4 = 50$ and $\lambda_5 = 100$. These are generated by ten calls to G05MEF, after initialization by G05KBF.

9.1 Program Text

```
*     G05MEF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT, M, N
      PARAMETER        (NOUT=6,M=5,N=10)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, IGEN, J
*     .. Local Arrays ..
      DOUBLE PRECISION VLAMDA(M)
      INTEGER          ISEED(4), X(M)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05MEF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05MEF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameter LAMBDA
      VLAMDA(1) = 0.5D0
      VLAMDA(2) = 5.0D0
      VLAMDA(3) = 1.0D1
      VLAMDA(4) = 5.0D2
      VLAMDA(5) = 1.0D3
*     Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 423442
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*     Generate integers and store in X
      DO 20 I = 1, N
         IFAIL = 1
         CALL G05MEF(M,VLAMDA,X,IGEN,ISEED,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
*
            WRITE (NOUT,99999) (X(J),J=1,M)
         ELSE
            WRITE (NOUT,99998) IFAIL
            GO TO 40
         END IF
   20 CONTINUE
*
   40 CONTINUE
*
99999 FORMAT (1X,5(1X,I12))
99998 FORMAT (1X,' ** G05MEF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05MEF Example Program Results

           1            3           13          482         1001
           1            4           12          494         1046
           1            2           11          548          941
           0            6            8          518          977
           0            6            8          504          943
           1            6            8          502          991
           0           11            7          475          991
           1            4            5          507         1012
           0            4           13          537         1016
           0            4            7          492         1072
```